Laboratorní cvičení - Diferenciální počet funkcí jedné reálné proměnné

Funkce jedné reálné proměnné

Elementární funkce a operace s funkcemi

Ukažme si nejprve jak se v Maple zapisují základní matematické operace sčítání, odčítání, násobení a dělení.Pro začátek lze využít pomůcky z lišty v paletě **Expression.** Jinak v Maplovském zápisníku používáme pro sčítání a odčítání znaky " $+$ " a " $-$ " z klávesnice, pro násobení je potřeba použít symbol " $*$ " a pro dělení symbol " / ". Pro psaní dalších symbolů je výhodné si otevřít paletu **Operators.**
Pro zápis některých elementárních funcí lze využít paletu **Expression** v levé liště, další zápisy funkcí lze nalézt v Heplu (Initially Known Mathematical Functions). Pro naše účely budou potřebné zápisy goniometrických, cyklometrických, logaritmických, exponenciálních a mocninných funkcí.
Syntaxe goniometrických funkcí:

```
> sin(x);cos(x);tan(x);cot(x);
```

$$\sin(x)$$

$$\cos(x)$$

$$\tan(x)$$

$$\cot(x)$$

Syntaxe cyklometrických funkcí (inverzní ke goniometrickým):

```
> arcsin(x);arccos(x);arctan(x);arccot(x);
```

$$\arcsin(x)$$

$$\arccos(x)$$

$$\arctan(x)$$

$$\operatorname{arccot}(x)$$

Syntaxe logaritmických a exponenciálních funkcí:

```
> ln(x);log(x);log10(x);log[b](x);exp(x);
```

$$\ln(x)$$

$$\ln(x)$$

$$\frac{\ln(x)}{\ln(10)}$$

$$\frac{\ln(x)}{\ln(b)}$$

$$\mathrm{e}^{x}$$

Mocnina a odmocnina:

```
> x^2;sqrt(x);root[2](x);root[5](x);
```

$$x^2$$

$$\sqrt{x}$$

$$\sqrt{x}$$
$$x^{1/5}$$

Zápis druhé mocniny funkce sinus (a dalších) se píše následovně:

```
> sin(x)^2;
```

$$\sin(x)^2$$

Všimněte si jiného zápisu pro odmocniny vyšších řádů, jsou zapsány ve tvaru mocninné funkce.

Příklad. Zkuste si napsat $\sin^2 x + 2 \cdot e^x - \frac{8}{9}$.

Definice vlastních funkcí

Vlastní funkci lze definovat různými způsoby, základní syntaxe je ***jméno_funkce:=proměnná->předpis***.

Příklad 1. Definujte funkci $f(x) = \frac{x^2 - 3}{5\,x}$.

```
> f:=x->(x^2-3)/(5*x);
```

$$f := x \rightarrow \frac{1}{5}\,\frac{x^2 - 3}{x}$$

Další možností, jak definovat funkci je využití nějakého předchozího zápisu výrazu, ze kterého pomocí příkazu ***unapply*****(*****výraz, proměnná*****)** lze nadefinovat funkci.

```
> a:=sin(x)/x;
```

$$a := \frac{\sin(x)}{x}$$

```
> f:=unapply(a,x);
```

$$f := x \rightarrow \frac{\sin(x)}{x}$$

Funkci lze také definovat pomocí procedury. Touto možnost ale vynecháme.
Jestliže je funkce definovaná po částech, použijeme pro její zadání příkaz ***jméno_funkce:=proměnná->piecewise(podm1, hod1, podm2, hod2,..., jinak)*** , kde "podm" jsou podmínky a "hod" jsou hodnoty funkce za dané podmínky a "jinak" je hodnota funkce, jestliže není splněna žádná z podmínek.

Příklad 2. Definujte funkci, která je zadána předpisem $f(x) = \begin{cases} 3\,x + 2 & x \le 1 \\ 6 - x & x > 1 \end{cases}$.

```
> f:=x->piecewise(x<=1,3*x+2,x>1,6-x);
```

$$f := x \rightarrow piecewise(x \le 1, 3\,x + 2, 1 < x, 6 - x)$$

```
> plot(f(x),x=-1..2);
```



Příklad 3. Definujte funkci, která je zadána předpisem $f(x) = \begin{cases} \sin x & 0 \le x \le 2\pi \\ 0 & jinak \end{cases}$.

```
> f:=x->piecewise(0<=x and x<=2*Pi,sin(x),0);
```

$$f := x \to piecewise\left(0 \le x \textbf{ and } x \le 2\pi, \sin(x), 0\right)$$

```
> plot(f(x),x=-Pi..3*Pi);
```



Vyhodnocování funkcí

Všechy funkce jsou definované v oboru komplexnch čísel, proto jsou např. přípustné i záporné hodnoty v logaritmu nebo v sudých odmocninách.

```
> ln(-5);
```

$$\ln(5) + I\pi$$

```
> sq rt(-25);
```

$$5\,I$$

V případě, že se nám ve výsledku objeví symbol **I**, značí to, že jsme někde udělali chybu v předpokladu o definičním oboru dané funkce, pokud ovšem opravdu nepracujeme s komplexními čísly.

Pro numerické vyčíslení výrazů obsahujících konstanty používáme příkazu **`evalf`** `(výraz)`.

```
> evalf(sin(2));
```

$$0.9092974268$$

```
> evalf(ln(-5));
```

$$1.609437912 + 3.141592654\,I$$

```
> evalf(Pi);
```

$$3.141592654$$

Pokud chceme získat výsledek na určitý počet cifer, připíšeme toto číslo do hranaté závorky za příkazu **`evalf`** .

```
> evalf[3](Pi);
```

$$3.14$$

```
> evalf[15](exp(1));
```

$$2.71828182845905$$

```
> evalf[8](cos(Pi/4)+sin(Pi/2));
```

$$1.7071068$$

Vyhodnotit výrazy obsahující konstanty lze i bez použití příkazu **`evalf`** , v tomto případě je potřeba zadat konstantu ve formě desetinného čísla.

```
> sin(2.);
```

$$0.9092974268$$

Pro nalezení funkční hodnoty zadané funkce v daném bodě postupujeme následovně:
1. funkci nadefinujeme - ***jméno_funkce:=proměnná-> předpis***,
. funkci vyhodnotí ***jméno_funkce(bod).***

Příklad 1. Vypočtěte funkční hodnotu funkce $f(x) = x^3 + 2x - 5$ v bodech $x_1 = 2, x_2 = -3$.

```
> f:=x->x^3+2*x-5;
```

$$f := x \to x^3 + 2x - 5$$

```
> f(2);
```

$$7$$

```
> f(-3);
```

$$-38$$

```
> f(1.25);
```

$$-0.546875$$

Graf funkce jedné proměnné

Nejjednodušší příkaz pro vykreslení grafu funkce je příkaz $\mathbf{plot}(\mathbf{funkce})$. Pokud nezadáme interval, na kterém chceme funkci vykreslit, bude výsledek zobrazen na intervalu $(-10, 10)$.

Příklad 1. Nakreslete graf funkce $f(x) = 3x - 2$.

```
> plot(3*x-2);
```



Pro vykreslení funkce na daném intervalu použijeme příkaz **plot**$(\mathbf{f}, \mathbf{x} = \mathbf{x}_0 .. \mathbf{x}_1)$.

Příklad 2. Nakreslete graf funkce $y = \tan x, x \in \langle -2\pi, 2\pi \rangle$.

```
> plot(tan(x),x=-2*Pi..2*Pi);
```



Jestliže zadaná funkce není na zobrazovaném intervalu ohraničená, projeví se nedostatky programu v tom, že použije příliš velký rozsah hodnot na ose *y*. Tento nedostatek lze potlačit tím, že sami nastavíme kromě rozzsahu na ose *x* i rozsah na ose *y*.

```
> plot(tan(x),x=-2*Pi..2*Pi,y=-2..2);
```



Další problém nastane, je-li funkce nespojitá. V tomto případě zůstávají v grafu navíc svislé čáry. Ty lze odstranit volbou **discont** = **true** .

```
> plot(tan(x),x=-2*Pi..2*Pi,y=-2..2,discont=true);
```



V případě, že chceme mít na osách stejné měřítko, použijeme příkaz **plot**$\left(\mathbf{f},\ \mathbf{x}=\mathbf{x}_0\,..\,\mathbf{x}_1,\ \mathbf{scaling}=\mathbf{constrained}\right)$. Srovnejte v následujícím příkladě.

Příklad 3. Nakreslete graf funkce $y=\cos x, x\in\langle -2\pi, 2\pi\rangle$.

```
> plot(cos(x),x=-2*Pi..2*Pi);
```




```
> plot(cos(x),x=-2*Pi..2*Pi,scaling=constrained);
```



Pomocí volby **tickmarks=[m, n]**lze nastavit počet dělících bodů na osách *x* a *y*,tento počet je pouze přibližný, Maple sám ošetří dělení intervalu, aby vypadalo "hezky". V případě, že bychom chtěli do obrázku umístit nadpis vztahující se ke grafu funkce, použijeme volbu **title=t,** kde t libovolný text ve dvojitých uvozovkách (tzv. řetězec).
Další skupina příkazů se bude týkat **tloušťky, barvy a stylu čar**. Tloušťku čáry měníme pomocí volby **thickness=n,** kde n je celé nezáporné číslo, defaultně je nastaveno thicness=0. Barvu čáry měníme volbou **color=název,** názvy jsou v angličině, přednastaven je color=red. Styl čáry měníme pomocí **linestyle=t,** kde t může být solid, dot, dash, dashdot, longdash, spacedash, spacedot. Defaultně je t solid. Styl čáry lze měnit také přiřazením čísla 1 až 7, která korespondují se zadanými názvy.Pokud chceme graf nakreslit pouze tečkovaně, použijeme nastavení **style=point.** Veškeré další informace lze nalézt v Helpu.
Příklad 4. Nakreslete graf funkce $y = e^x$,$x \in \langle -3, 3\rangle$ tak, aby graf byl naznačen pouze tečkovaně a měl zelenou barvu, na ose *x* bylo 5 dělících bodů a na ose *y*4 a nadpis byl Exponenciální funkce..

```
> plot(exp(x),x=-3..3,color=green,style=point,
tickmarks=[5,4],title="Exponenciální funkce");
```



Pro kreslení dvou a více funkcí do jednoho obrázku je potřeba použít příkaz **plot([f1,f2,...], x=x0..x1).**

Příklad 5. Nakreslete do jednoho obrázku grafy funkcí $y = \frac{1}{x}, y = e^{-x}$, a to tak, aby první graf byl modrou barvou a druhý zelenou, tloušťku čar zvolte 2. Až graf nakreslíte, upravte vhodně měřítko na ose *y*a odstraňte nespojitosti v grafu .

```
> plot([1/x,exp(-x)],x=-5..5,color=[blue, green], thickness=2);
```



Odstraňte všechny nedostatky v obrázku, tj. upravte vhodně měřítko na ose *y* a odstraňte nespojitosti v grafu funkce $y = \frac{1}{x}$.

```
> plot([1/x,exp(-x)],x=-5..5,y=-2..2,color=[blue, green],
thickness=2, discont=true);
```



Pozn. Graf funkce lze použít i pro odhad definičního oboru funkce tak, že zvolíme velký rozsah hodnot *x*a graf se vykreslí pouze pro za *x*, pro která je funkce definována.

Příklad 6. Určete definiční obor funkce $y = \sqrt{x}$.

```
> plot(sqrt(x), x=-4..4);
```



Limita funkcí jedné proměnné

Limitu lze vypočítat použitím příkazu `limit(f, x = a, dir)` , kde f je funkce, $x$je proměnná, a je bod, ve kterém limitu počítáme a dir je nepovinný parametr, pro limitu zprava je dir = right a pro limitu zleva je dir = left. Pokud použijeme zápis `Limit(f, x = a, dir)` , výpočet se neprovede, jen se formálně zapíše.

Oboustranná limita

Příklad 1. Vypočtěte limitu $\lim_{x \to 2} \frac{x-3}{x^2-9}$.

```
> f:=(x-3)/(x^2-9);
```

$$f := \frac{x-3}{x^2-9}$$

```
> limit(f,x=2);
```

$$\frac{1}{5}$$

Funkci $f(x)$ není třeba definovat, ale lze psát přímo:

```
> limit((x-3)/(x^2-9),x=2);
```

$$\frac{1}{5}$$

Příklad 2. Zapište a vypočtěte limitu $\lim_{x \to 1} (\ln x + x^2 + 3)$.

```
> Limit(ln(x)+x^2+3,x=1)=limit(ln(x)+x^2+3,x=1);
```

$$\lim_{x \to 1} \left(\ln(x) + x^2 + 3\right) = 4$$

Příklad 3. Zapište a vypočtěte limitu $\lim\limits_{x \to \infty} \frac{\ln x}{x}$.

```
> Limit(ln(x)/x,x=infinity)=limit(ln(x)/x,x=infinity);
```

$$\lim_{x \to \infty} \frac{\ln(x)}{x} = 0$$

V případě, že limita neexistuje, lze získat dva typy výsledků:
. výde interval, což žfunkce v blíécilujíními intervalu.
. "undefined", v tomto přpotřt vý jednostrannýterébý hodnoty.

Příklad 4. Vypočtěte limitu $\lim\limits_{x \to 0} \sin\frac{1}{x}$.

```
> limit(sin(1/x),x=0);
```

$$-1..1$$

Představu o tom, jak vypadá graf funkce v okolí bodu x =0, získáme pomocí obrázku.

```
> plot(sin(1/x),x=-1..1);
```



Příklad 5. Vypočtěte limitu $\lim\limits_{x \to 0} \frac{\sin x + 1}{\sin x}$.

```
> limit((sin(x)+1)/sin(x),x=0);
```

undefined

Pozor! V případě, že výsledek limity je ∞, $-\infty$, pak limita dané funkce existuje, ale je nevlastní.

Příklad 6. Vypočtěte limitu $\lim_{x \to 0} \frac{\cos x + 1}{\cos x - 1}$.

```
> limit((cos(x)+1)/(cos(x)-1),x=0);
```

$$-\infty$$

Příklad 7. Vypočtěte limitu $\lim_{x \to 0} \frac{1}{x^2}$.

```
> limit(1/x^2,x=0);
```

$$\infty$$

Jednostranné limity

V případě, že nám Maple dává výsledek limity "undefined", víme, že limita v daném bodě neexistuje. Můžeme, ale spočítat tzv. jednostranné limity.

Příklad 1. Vypočtěte limitu $\lim_{x \to 0} \frac{\sin x + 1}{\sin x}$.

```
> limit((sin(x)+1)/sin(x),x=0);
```

$$\mathit{undefined}$$

```
> limit((sin(x)+1)/sin(x),x=0,right);
```

$$\infty$$

```
> limit((sin(x)+1)/sin(x),x=0,left);
```

$$-\infty$$

Příklad 2. Vypočtěte limitu $\lim_{x \to 1^+} \frac{|x^2 - 1|}{x - 1}$.

```
> limit(abs(x^2-1)/(x-1),x=1,right);
```

$$2$$

Příklad 3. Vypočtěte limitu $\lim_{x \to 1^-} \frac{|x^2 - 1|}{x - 1}$.

```
> limit(abs(x^2-1)/(x-1),x=1,left);
```

$$-2$$

Příklad 4. Vypočtěte limitu $\lim_{x \to \frac{\pi}{2}^-} \operatorname{tg} x$.

```
> limit(tan(x),x=Pi/2,left);
```

$$\infty$$

V případě funkcí definovaných po částech lze jednostranné limity počítat následovně.

Příklad 5. Nalezněte limity $\lim_{x \to 3^+} f(x)$ a $\lim_{x \to 3^-} f(x)$, jestliže funkce $f(x)$ je definována následovně

$$f(x) = \begin{cases} x + 1 & 3 < x \\ x^2 - 1 & \mathit{otherwise.} \end{cases}$$

```
> f:=piecewise(x>3,x+1,x^2-1);
```

$$f := \begin{cases} x + 1 & 3 < x \\ x^2 - 1 & \mathit{otherwise} \end{cases}$$

```
> limit(f,x=3,right);
```

$$4$$

```
> limit(f,x=3,left);
```

$$8$$

Derivace funkcí jedné proměnné

Derivace funkcí daných explicitně

Derivace funkcí

Pro výpočet derivace můžeme použít příkaz $\frac{d}{dx}f(x) =$ `diff(f(x), x)` .

Příklad 1. Vypočtěte derivaci funkce $f(x) = \ln(1 + \cos x)$.

```
> f:=ln(1+cos(x));
```

$$f := \ln(\cos(x) + 1)$$

```
> diff(f,x);
```

$$-\frac{\sin(x)}{\cos(x) + 1}$$

Tyto dva kroky lze napsat jedním příkazem.

```
> diff(ln(1+cos(x)),x);
```

$$-\frac{\sin(x)}{\cos(x) + 1}$$

Výsledek, který takto získáme, je výraz
. Pokud chceme s výsledkem dále pracovat jako s funcí,

je potřeba použít jiného příkazu pro výpočet derivace, a to $\frac{d}{dx}f(x) =$ `D(f)(x)` . V tomto případě je potřeba nejprve nadefinovat funkci f jako funkci proměnné x .

Příklad 2. Vypočtěte derivaci funkce $f(x) = 1 + 5\cos x$.

```
> f:=x->1+5*cos(x);
```

$$f := x \to 1 + 5\cos(x)$$

```
> D(f);
```

$$x \to -5\sin(x)$$

Symbolický zápis pro derivaci je možné získat pomocí příkazu $\frac{d}{dx}f(x) =$ `Diff(f(x), x)` . pomocí tohoto příkazu se derivace nepočítá, pouze se zapíše.

Příklad 3. Zapište symbolicky a vypočtěte derivaci funkce $f(x) = \cos \pi x - \sin 2x$.

```
> f:=cos(Pi*x)-sin(2*x);
```

$$f := \cos(\pi x) - \sin(2x)$$

```
> Diff(f,x)=diff(f,x);
```

$$\frac{d}{dx}\left(\cos(\pi x) - \sin(2x)\right) = -\sin(\pi x)\,\pi - 2\cos(2x)$$

Derivace vyšších řádů

Pro výpočet druhých derivací lze použít příkazů `diff(f(x), x, x)` nebo `D(D(f))(x)` .

Příklad 1. Vypočtěte druhou derivaci funkce $f(x) = x^4 + 5x^3 - 2x$.

```
> diff(x^4+5*x^3-2*x,x,x);
```

$$12x^2 + 30x$$

Příklad 2. Vypočtěte druhou derivaci funkce $f(x) = \ln(x^2 + 1)$.

```
> f:=x->ln(x^2+1);
```

$$f := x \rightarrow \ln(x^2 + 1)$$

```
> D(D(f))(u);
```

$$\frac{2}{u^2 + 1} - \frac{4\,u^2}{(u^2 + 1)^2}$$

Derivací vyšších řádů zapisujeme stručněji použiítím příkazů `diff(f(x),x$n)` nebo `(D@@n)(f)(x)` .

Příklad 3. Vypočtěte třetí derivaci funkce $f(x) = xe^{2x}$.

```
> diff(x*exp(2*x),x$3);
```

$$12\,e^{2x} + 8\,x\,e^{2x}$$

Příklad 4. Vypočtěte pátou derivaci funkce $f(x) = x^6 + \ln x$.

```
> f:=x->x^6+ln(x);
```

$$f := x \rightarrow x^6 + \ln(x)$$

```
> (D@@5)(f)(x);
```

$$720\,x + \frac{24}{x^5}$$

Příklad 5. Vypočtěte *n*-tou derivaci funkce $f(x) = \ln x$.

```
> diff(ln(x),x$n);
```

$$\frac{(-1 + x)^{1 - n}\,\mathrm{hypergeom}([1, 1], [2 - n], 1 - x)}{\Gamma(2 - n)}$$

Příklad 6. Zapište a vypočtěte druhou derivaci funkce $f(x) = \frac{\sin x}{x}$.

```
> f:=sin(x)/x;
```

$$f := \frac{\sin(x)}{x}$$

```
> Diff(f,x$2)=diff(f,x,x);
```

$$\frac{d^2}{dx^2}\left(\frac{\sin(x)}{x}\right) = -\frac{\sin(x)}{x} - \frac{2\cos(x)}{x^2} + \frac{2\sin(x)}{x^3}$$

Derivace funkcí v daném bodě

Jestliže máme vypočítat derivaci funkce v daném bodě, je potřeba definovat derivaci jako funkci, tzn. že pro výpočet derivace je potřeba použít operátoru $\frac{d}{dx} f(x) =$ `D(f)(x)` .

Příklad 1. Vypočtěte derivaci funkce $f(\mathrm{x}) = \mathrm{x}^4 - 3\,\mathrm{x}^2 + 2$ v bodě $\mathrm{x}_0 = 0, \mathrm{x}_1 = 1$.

```
> f:=x->x^4-3*x^2+2;
```

$$f := x \rightarrow x^4 - 3\,x^2 + 2$$

```
> D(f)(0);
```

$$0$$

```
> D(f)(1);
```

$$-2$$

Příklad 2. Vypočtěte druhou derivaci funkce $f(\mathrm{x}) = \frac{x^3}{x^2 - 1}$ v bodě $\mathrm{x}_0 = 2$.

```
> f:=x->x^3/(x^2-1);
```

$$f := x \to \frac{x^3}{x^2 - 1}$$

```
> (D@@2)(f)(x);
```

$$\frac{6\,x}{x^2 - 1} - \frac{14\,x^3}{\left(x^2 - 1\right)^2} + \frac{8\,x^5}{\left(x^2 - 1\right)^3}$$

```
> (D@@2)(f)(2);
```

$$\frac{28}{27}$$

Inverzní funkce

Řešení výpočtem

Příklad 1. Je dána funkce $y = \sin(3\,x - 1)$. Určete interval, na kterém je funkce prostá, nalezněte funkci inverzní a obě funkce vykreslete do jednoho obrázku.

```
> solve(3*x-1=-Pi/2);
```

$$-\frac{1}{6}\,\pi + \frac{1}{3}$$

```
> solve(3*x-1=Pi/2);
```

$$\frac{1}{6}\,\pi + \frac{1}{3}$$

Funkce je prostá na intervalu $\left\langle \frac{1}{3} - \frac{\pi}{6}, \frac{1}{3} + \frac{\pi}{6} \right\rangle$.

Určíme funkci inverzní k dané funkci, opět pomocí příkazu **solve(rovnice, proměnná),**kde proměnná je v našem případě *x*, protože tu se snažíme z rovnice vyjádřit.

```
> solve(sin(3*x-1)=y,x);
```

$$\frac{1}{3} + \frac{1}{3}\arcsin(y)$$

Nyní zakreslíme obě funkce a osu 1. a 3. kvadrantu do jednoho obrázku. K tomu potřebujeme načíst balíček **plots.** Ještě před vykreslením je potřeba v inverzní funkci zaměnit proměnné *x* a y.

```
> with(plots):
> p1:=plot(sin(3*x-1),x=1/3-Pi/6..1/3+Pi/6):
> p2:=plot(1/3+1/3*arcsin(x),x=-1..1,color=blue):
> p3:=plot(x,x=-1..1,color=green):
> display({p1,p2,p3},scaling=constrained);
```



Řešení pomocí zabudované funkce

Pro nalezení inverzní funkce k dané funkci je potřeba načít pomocí příkazu **with**balíček **Student[Calculus1]**, v tomto balíčku je zabudovaná funkce **InversePlot(f(x), x = a..b),** která do jednoho obrázku vykreslí zadanou funkci a k ní funkci inverzní.

Příklad 1. K funkci $y = e^x$ nalezněte funkci inverzní.

Vzhledem k tomu, že exponenciální funkce je prostá na celém svém definičním oboru, můžeme si zadat libovolný interval, na kterém funkci a funkci k ní inverzní zobrazíme.

```
> with(Student[Calculus1]):
> InversePlot(exp(x),x=-1..1);
```



Příklad 2. K funkci $y = 1 + \mathrm{arctg}(3x - 4)$ najděte funkci inverzní.
POZOR! Funkce kosinus není prostá na celém definičním, pokud použijeme příkazu **InversePlot(f(x), x = a..b**) bez správného určení intervalu, kde je zadaná funkce prostá, dostaneme sice výsledek, ale ten není správný.

```
> InversePlot(2-cos(2*x+1),x=-5..5);
```



Příklad 3. Najděte funkci inverzní k funkci $y = \sqrt{1+e^{2x}}$, danou funkci i funkci inverzní vykreslete.

```
> plot(sqrt(1+exp(2*x)));
```



Definiční obor je $\mathbb{R}$. Inverzní funkci získáme pomocí příkazu **solve**.

```
> solve(sqrt(1+exp(2*x))=y,x);
```

$$\frac{1}{2}\ln\left(-1+y^2\right)$$

Pokud si necháme vykreslit, jak vypadá graf inverzní funkce zjistíme, že řešením jsou dvě větve logaritmu, což však vyhledem k původní funkci být nemůže. V našem případě je řešením pouze část pro kladné hodnoty *x*.

```
> plot((1/2)*ln(-1+x^2));
```



Pokud použijeme zabudovanou funkci, do problémů s druhou větví logaritmu se nedostaneme, program sám tuto variantu ošetří.

```
> InversePlot(sqrt(1+exp(2*x)),x=-1..1);
```


The Inverse of
f(x) = (1+exp(2*x))^(1/2)
3
2
1
-1
0
1
2
3
x
-1
f(x)
The inverse of f(x)

>